
(%i43) kill(all)$
assume(R1>0,R2>0,R3>0,Rg>0,Rz>0,R>0)$

Výpočet útlumového článku Pí + T

Výpočet je proveden v CAS programu wxMaxima [http://maxima.sourceforge.net/]
[http://en.wikipedia.org/wiki/List_of_computer_algebra_systems]
Vypracoval: Derka Ladislav

1 Článek T - Výpočet odporu zdroje a zátěže

Součástky jsou zapojeny postupně od zdroje k zátěži:

Figure 1: /media/Data/wxMaxima/Elektro/Att_T.png

1.1 Odvození vzorců - směr Rz -> Rg

Jedna rovnice

(%i2) RaT: (Rz+R3)$
eqaT: Rg=R1+RaT*R2/(RaT+R2);

1.2 Odvození vzorců - směr Rg -> Rz

Druhá rovnice

(%i4) RbT: (Rg+R1)$
eqbT: Rz=R3+RbT*R2/(RbT+R2);

1.3 Výpočet odporu Rg + Rz obecně

Zadání známých hodnot [R1,R2,R3]

(%i6) ZadaniT: [R1=150,R2=300,R3=150];

Soustava dvou rovnic o dvou neznámých [Rg,Rz]

(%i7) eqT: at([eqaT,eqbT],ZadaniT);

Řešení pomocí solveru (v excelu lze použít řešitel)

(%i8) soT: solve(eqT,[Rg,Rz]);
srT: float(soT[2]);

1.4 Výpočet útlumu (pokračování)

Sestavení obecné rovnice článku T pro útlum [dB]

(%i10) RcT: (R2*(R3+Rz))/(R3+R2+Rz)$
UtT: 20*log(2*((R1+RcT)/(Rg+RcT+R1))*(RcT/(RcT+R1))*(Rz/(Rz+R3)))/log(10);

Automatické dosazení a výsledek v dB

(%i12) float(at(at(UtT,ZadaniT),srT));

1.5 Výpočet odporu Rg + Rz pokud R1 = R3

Pokud R1=R3, pak obvod je symetrický a platí Rg=Rz
Pro jednoduchost označíme Rg a Rz symbolem R
Počítáme tedy R z jedné rovnice (z druhé vyjde stejný výsledek)

(%i13) at(eqaT,[R3=R1,Rg=R,Rz=R]);
solve(%,R);
sjT: %[2];

Dosadíme známé hodnoty a vyjde číselně R

(%i16) float(at(sjT, ZadaniT));

1.6 Výpočet R1,R2 z požadovaného útlumu

Symetrický článek T:
platí R3 = R1
R je vstupní i výstupní impedance

(%i17) k = 10^(-abs(dB)/20);
R1 = R * (1 - k)/(1 + k);
R2 = R * 2/(1/k - k);

Zapsáno jako funkce

(%i20) fT(R,dB):=block([k,R1,R2],
    k: 10^(-abs(dB)/20),
    R1: R * (1 - k)/(1 + k),
    R2: R * 2/(1/k - k),
    [R1,R2]
)$

Kontrolní výpočet

(%i21) [R1,R2] = fT(335.4101966249684, 8.359505609999157);

2 Článek Pí - Výpočet odporu zdroje a zátěže

Součástky jsou zapojeny postupně od zdroje k zátěži:

Figure 2: /media/Data/wxMaxima/Elektro/Att_Pi.png

2.1 Odvození vzorců - směr Rz -> Rg

Jedna rovnice

(%i22) RaP: Rz*R3/(Rz+R3)+R2$
eqaP: Rg=RaP*R1/(RaP+R1);

2.2 Odvození vzorců - směr Rg -> Rz

Druhá rovnice

(%i24) RbP: Rg*R1/(Rg+R1)+R2$
eqbP: Rz=RbP*R3/(RbP+R3);

2.3 Výpočet odporu Rg + Rz

Zadání známých hodnot [R1,R2,R3]

(%i26) ZadaniP: [R1=200,R2=60,R3=200];

Soustava dvou rovnic o dvou neznámých [Rg,Rz]

(%i27) eqP: at([eqaP,eqbP],ZadaniP);

Řešení pomocí solveru (v excelu lze použít řešitel)

(%i28) soP: solve(eqP,[Rg,Rz]);

srP: float(%)[2];

2.4 Výpočet útlumu (pokračování)

Sestavení obecné rovnice článku Pí pro útlum [dB]

(%i30) RcP: (R3*Rz)/(R3+Rz)$
RdP: (RcP+R2)*R1/(RcP+R2+R1)$
UtP: 20*log(2*(RdP/(RdP+Rg))*(RcP/(R2+RcP)))/log(10);

Automatické dosazení a výsledek v dB

(%i33) float(at(at(UtP,ZadaniP),srP));

2.5 Výpočet odporu Rg + Rz pokud R1 = R3

Pokud R1=R3, pak obvod je symetrický a platí Rg=Rz
Pro jednoduchost označíme Rg a Rz symbolem R
Počítáme tedy R z jedné rovnice (z druhé vyjde stejný výsledek)

(%i34) at(eqaP,[R3=R1,Rg=R,Rz=R]);
solve(%,R);
sjP: %[2];

Dosadíme známé hodnoty a vyjde číselně R

(%i37) float(at(sjP, ZadaniP));

2.6 Výpočet R1,R2 z požadovaného útlumu

Symetrický článek Pí:
platí R3 = R1
R je vstupní i výstupní impedance

(%i38) k = 10^(-abs(dB)/20);
R1 = R * (1 + k)/(1 - k);
R2 = R * (1/k - k)/2;

Zapsáno jako funkce

(%i41) fP(R,dB):=block([k,R1,R2],
    k: 10^(-abs(dB)/20),
    R1: R * (1 + k)/(1 - k),
    R2: R * (1/k - k)/2,
    [R1,R2]
);

Kontrolní výpočet

(%i42) [R1,R2] = fP(72.23151185146152, 6.570292782303837);

Created with wxMaxima.

